
Corrections required for IPv6

Change 1

Method name
f_EUTRA_Preamble_Reject

Reason for change
Missing handling of IP user plane data in test cases using f_EUTRA_Preamble_Reject

Summary of change
Added handling of IP user plane data

Source of change
EUTRA_NASCommonFunctions.ttcn

Before change:

  function f_EUTRA_Preamble_Reject(CellId_Type p_CellId,

                                   NAS_CauseValue_Type p_RejectCause,

                                   EUTRA_ATTACH_Type p_ForcedAttach) runs on EUTRA_PTC

  {

    // First switch cell back on

    f_EUTRA_SetCellPower ( p_CellId, tsc_ServingCellRS_EPRE );

    f_EUTRA_AttachReject (p_CellId, p_RejectCause, p_ForcedAttach); // @sic R5s110176 sic@

    // And then just need to switch UE off

    f_UT_SwitchOffUE (UT, false, CNF_REQUIRED); // @sic R5s1100395 sic@

  }

After change:

  function f_EUTRA_Preamble_Reject(CellId_Type p_CellId,

                                   NAS_CauseValue_Type p_RejectCause,

                                   EUTRA_ATTACH_Type p_ForcedAttach) runs on EUTRA_PTC

  {

    // First switch cell back on

    f_EUTRA_SetCellPower ( p_CellId, tsc_ServingCellRS_EPRE );

    // Activate DHCP and ICMPv6

    f_DHCPv4_Start(IP, px_IPv4_RemoteAddress, cs_DrbInfo_EUTRA(p_CellId, tsc_DRB1));   /* @sic R5s100135, R5s100189 sic@ */

    f_ICMPv6_Start(IP, px_IPv6_RemoteAddress, cs_DrbInfo_EUTRA(p_CellId, tsc_DRB1));   /* @sic R5-106675 IP and IMS support sic@ */
    f_EUTRA_AttachReject (p_CellId, p_RejectCause, p_ForcedAttach); // @sic R5s110176 sic@

    // And then just need to switch UE off

    f_UT_SwitchOffUE (UT, false, CNF_REQUIRED); // @sic R5s1100395 sic@

  }

Change 2

Method name
f_EUTRA_IdleUpdated_Step15_17, f_EUTRA_NAS_AttachComplete

Reason for change
Delay is missing to expect ICMP signalling in test cases which uses this function in the test case body.

Summary of change
Removed delay from the common function f_EUTRA_IdleUpdated_Step15_17.
Added delay to the common function f_EUTRA_NAS_AttachComplete.

Source of change
EUTRA_CommonProcedures.ttcn, EUTRA_NASSteps.ttcn

Before change:

// EUTRA_CommonProcedures.ttcn

  function f_EUTRA_IdleUpdated_Step15_17 (CellId_Type p_CellId,

                                          boolean p_ReleaseConnection := true) runs on EUTRA_PTC

  { /* @sic R5-112600: always delay of 1.2s sic@ */

    var HalfOctet_Type v_EpsBearerId := tsc_EpsDefaultBearerId;

    f_EUTRA_NAS_AttachComplete(p_CellId, v_EpsBearerId);                        // step 16

    f_Delay(f_EUTRA_SetTimerToleranceMax(p_CellId, nonProtocolTimer, 1.2));     /* 1.2s delay acc. to 36.508 cl. 4.5A.1 to allow user-plane signalling */

    if (p_ReleaseConnection) { // @sic R5-105005 Some tests only want to do up to step 16 sic@

      f_EUTRA_RRC_ConnectionRelease(p_CellId);                                  // step 17

    }

  }

// EUTRA_NASSteps.ttcn

function f_EUTRA_NAS_AttachComplete(CellId_Type p_CellId,

                                      HalfOctet_Type p_EpsBearerId ) runs on EUTRA_PTC

  {

    SRB.receive(car_SRB2_NasPdu_IND(p_CellId,

                                    cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,

                                                                      cr_508_ATTACH_COMPLETE,

                                                                      cr_508_ActivateDefEpsBearerContextAccept(

                                                                                                               p_EpsBearerId))));

  };

After change:

// EUTRA_CommonProcedures.ttcn

  function f_EUTRA_IdleUpdated_Step15_17 (CellId_Type p_CellId,

                                          boolean p_ReleaseConnection := true) runs on EUTRA_PTC

  { /* @sic R5-112600: always delay of 1.2s sic@ */

    var HalfOctet_Type v_EpsBearerId := tsc_EpsDefaultBearerId;

    f_EUTRA_NAS_AttachComplete(p_CellId, v_EpsBearerId);                        // step 16

//    f_Delay(f_EUTRA_SetTimerToleranceMax(p_CellId, nonProtocolTimer, 1.2));     /* 1.2s delay acc. to 36.508 cl. 4.5A.1 to allow user-plane signalling */

    if (p_ReleaseConnection) { // @sic R5-105005 Some tests only want to do up to step 16 sic@

      f_EUTRA_RRC_ConnectionRelease(p_CellId);                                  // step 17

    }

  }

// EUTRA_NASSteps.ttcn

   function f_EUTRA_NAS_AttachComplete(CellId_Type p_CellId,

                                      HalfOctet_Type p_EpsBearerId

                                      ) runs on EUTRA_PTC

  {

    SRB.receive(car_SRB2_NasPdu_IND(p_CellId,

                                    cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,

                                                                      cr_508_ATTACH_COMPLETE,

                                                                      cr_508_ActivateDefEpsBearerContextAccept(

                                                                                                               p_EpsBearerId))));

    f_Delay(f_EUTRA_SetTimerToleranceMax(p_CellId, nonProtocolTimer, 1.2));     /* 1.2s delay acc. to 36.508 cl. 4.5A.1 to allow user-plane signalling */
  };

Change 3

Test case name
9_2_1_1_1

Reason for change
Missing delay in certain test cases, where template cr_508_ATTACH_COMPLETE is used, to allow IP user plane data handling.

Similar changes required in:

9_2_1_1_1a, 9_2_1_1_2. 9_2_1_2_1b, 9_2_1_1_19, 9_2_1_1_26, 9_2_1_2_3, 9_2_1_2_1b

Summary of change
Added delay to the common function and used it here to ensure TTCN waits for the ICMP signalling.

Source of change
NAS_Attach_EOnly.ttcn

Before change:

function fl_TC_9_2_1_1_1_Body(GutiParameters_Type p_GutiParams,

                                GutiParameters_Type p_GutiOldParams) runs on EUTRA_PTC

  {

    // ==========================================================================

    // initialize constant test case data structures

    // ==========================================================================

    var HalfOctet_Type v_EPS_Bearer_Id := tsc_EpsDefaultBearerId;

    var ProcedureTransactionIdentifier v_EPS_TI := tsc_PTI_1;

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var template (omit) MobileIdentity v_MSId := f_GetMSId (EPS_Only);  // @sic R5s110176, R5-103847 sic@

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (p_GutiParams.PLMN_Identity);

    var NAS_PlmnId v_NAS_Old_PlmnId := f_Asn2Nas_PlmnId(p_GutiOldParams.PLMN_Identity);

    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode ( eutra_CellA );

    var template (omit) LocAreaId v_LAI := f_GetLAI(v_PLMN, v_LAC, EPS_Only); // @sic R5s110176 sic@

    //v_AdditionalUpdateType is also used to follow TTCN3 v411 standard

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType (EPS_Only);  // @sic R5s110176, R5-103681 sic@

    var template (value) AccessPointName v_APN;

    var template (omit) ProtocolConfigOptions v_Pco := omit;

    var NAS_AttDetValue_Type v_AttachType :=f_GetEAttachType(EPS_Only); // @sic R5s110176 sic@

    var SRB_COMMON_IND v_NasInd;

    var NAS_UL_Message_Type v_PdnConnectivityRequest;

    var boolean v_EIT_Flag := false;

    // Switch on UE

    //@siclog "Step 1" siclog@

    f_UT_SwitchOnUE (UT, false);

    //@siclog "Step 2" siclog@

    //+ Check: Does the UE transmit an RRCConnectionRequest message not including S-TMSI

    //         and with establishmentCause set to mo-Signalling?

    SRB.receive(car_SRB0_RrcPdu_IND(eutra_CellA,

                                    cr_RRCConnectionRequest_withUE_Identity(cr_InitialUE_Identity_Random, mo_Signalling)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    //@siclog "Step 3" siclog@

    // send RRC Connection Setup on SRB0 (CCCH):

    f_EUTRA_RRC_ConnectionSetup_Def (eutra_CellA);

    // UE shall establish SRB1 acc. to 36.331 cl. 5.3.3.4 and 5.3.9

    // UE enters RRC_CONNECTED state

    //@siclog "Step 4" siclog@

    // Check: does the UE transmit an RRCConnectionSetupComplete message including MME group ID and MME code

    // with ATTACH REQUEST message including a GUTI and a PDN CONNECTIVITY REQUEST message?

    SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellA,

                                       cr_508_RRCConnectionSetupComplete(v_RRC_TI,

                                                                         ?,

                                                                         cr_RegisteredMME(p_GutiOldParams.PLMN_Identity, // @sic R5s100008 sic@

                                                                                          p_GutiOldParams.MME_GroupId,

                                                                                          p_GutiOldParams.MME_Code)),

                                       cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,

                                                                         cdr_ATTACH_REQUEST_WithGuti(v_AttachType,

                                                                                                     v_AdditionalUpdateType,

                                                                                                     ?,

                                                                                                     f_GutiParameters2MobileIdentity(omit, p_GutiOldParams), // GUTI1

                                                                                                     cr_TAI( v_NAS_Old_PlmnId, tsc_TAC2)),

                                                                         cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, ?))))

      -> value v_NasInd;

    v_PdnConnectivityRequest := v_NasInd.Signalling.Nas[0].Pdu.PiggybackedPduList[0].Msg;

    v_EPS_TI := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier; // Use this value in SS responses

    // Get the PCO to be used later (TS 36.508 v.8.2.2 Table 4.7.3-6)

    if (ispresent( v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions)) {

      v_Pco := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions;

    }

    //+   The SS transmits an AUTHENTICATION REQUEST message to initiate the EPS

    //+   authentication and AKA procedure.

    //+   The UE transmits an AUTHENTICATION RESPONSE message and establishes

    //+   mutual authentication.

    //+   The SS transmits a NAS SECURITY MODE COMMAND message to

    //+   activate NAS security.

    //+   The UE transmits a NAS SECURITY MODE COMPLETE message

    //+   and establishes the initial security configuration.

    //+  - ESM information optionally transferred

    //@siclog "Step 5 - 9" siclog@

    v_EIT_Flag := f_CheckEsmInfoTransferFlag(v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST);

    v_APN := f_EUTRA_Authentication_NAS_AS_Security (eutra_CellA, v_EIT_Flag, v_EPS_TI, true);

    //@siclog "Step 10" siclog@

    //+   SS responds with ATTACH ACCEPT message including a valid TAI list. The

    //+   ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message is piggybacked in

    //+   ATTACH ACCEPT message.

    //+   Note: The IP addresses of the UE are not allocated in this test so PDN

    //+   address is not included in the message.

    f_EUTRA_RRCConnectionReconfigurationWithAttachAcceptAllParams(eutra_CellA,

                                                                  v_EPS_Bearer_Id,

                                                                  v_RRC_TI,

                                                                  v_EPS_TI,

                                                                  v_AttachType,

                                                                  f_EUTRA_GetPDNAddress(f_CheckPCOforIPallocationViaNas (v_Pco)),

                                                                  f_GutiParameters2MobileIdentity(tsc_IEI_Guti, p_GutiParams), // GUTI1

                                                                  cs_TrackingAreaIdListDef_lv,

                                                                  v_LAI,

                                                                  v_MSId,

                                                                  omit,

                                                                  v_APN,

                                                                  f_GetDefPCOforPCSCFAddress (v_Pco), // @sic R5-104703 sic@

                                                                  f_GetAdditionalUpdateResult(v_NasInd.Signalling.Nas[0].Pdu.Msg.aTTACH_REQUEST.addUpdateType,

                                                                                              EPS_Only));  // @sic R5s110176, R5-103681 sic@

    //@siclog "Step 11" siclog@

    //+   Check: does the UE transmit an ATTACH COMPLETE message including an

    //+   ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message?

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,

                                                                      cr_508_ATTACH_COMPLETE,

                                                                      cr_508_ActivateDefEpsBearerContextAccept(v_EPS_Bearer_Id))));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 11"); // TP2

    //@siclog "Step 12 Void" siclog@

    //@siclog "Step 13" siclog@

    //+   The SS releases the RRC connection.

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);

After change:

function fl_TC_9_2_1_1_1_Body(GutiParameters_Type p_GutiParams,

                                GutiParameters_Type p_GutiOldParams) runs on EUTRA_PTC

  {

    // ==========================================================================

    // initialize constant test case data structures

    // ==========================================================================

    var HalfOctet_Type v_EPS_Bearer_Id := tsc_EpsDefaultBearerId;

    var ProcedureTransactionIdentifier v_EPS_TI := tsc_PTI_1;

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var template (omit) MobileIdentity v_MSId := f_GetMSId (EPS_Only);  // @sic R5s110176, R5-103847 sic@

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (p_GutiParams.PLMN_Identity);

    var NAS_PlmnId v_NAS_Old_PlmnId := f_Asn2Nas_PlmnId(p_GutiOldParams.PLMN_Identity);

    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode ( eutra_CellA );

    var template (omit) LocAreaId v_LAI := f_GetLAI(v_PLMN, v_LAC, EPS_Only); // @sic R5s110176 sic@

    //v_AdditionalUpdateType is also used to follow TTCN3 v411 standard

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType (EPS_Only);  // @sic R5s110176, R5-103681 sic@

    var template (value) AccessPointName v_APN;

    var template (omit) ProtocolConfigOptions v_Pco := omit;

    var NAS_AttDetValue_Type v_AttachType :=f_GetEAttachType(EPS_Only); // @sic R5s110176 sic@

    var SRB_COMMON_IND v_NasInd;

    var NAS_UL_Message_Type v_PdnConnectivityRequest;

    var boolean v_EIT_Flag := false;

    // Switch on UE

    //@siclog "Step 1" siclog@

    f_UT_SwitchOnUE (UT, false);

    //@siclog "Step 2" siclog@

    //+ Check: Does the UE transmit an RRCConnectionRequest message not including S-TMSI

    //         and with establishmentCause set to mo-Signalling?

    SRB.receive(car_SRB0_RrcPdu_IND(eutra_CellA,

                                    cr_RRCConnectionRequest_withUE_Identity(cr_InitialUE_Identity_Random, mo_Signalling)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    //@siclog "Step 3" siclog@

    // send RRC Connection Setup on SRB0 (CCCH):

    f_EUTRA_RRC_ConnectionSetup_Def (eutra_CellA);

    // UE shall establish SRB1 acc. to 36.331 cl. 5.3.3.4 and 5.3.9

    // UE enters RRC_CONNECTED state

    //@siclog "Step 4" siclog@

    // Check: does the UE transmit an RRCConnectionSetupComplete message including MME group ID and MME code

    // with ATTACH REQUEST message including a GUTI and a PDN CONNECTIVITY REQUEST message?

    SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellA,

                                       cr_508_RRCConnectionSetupComplete(v_RRC_TI,

                                                                         ?,

                                                                         cr_RegisteredMME(p_GutiOldParams.PLMN_Identity, // @sic R5s100008 sic@

                                                                                          p_GutiOldParams.MME_GroupId,

                                                                                          p_GutiOldParams.MME_Code)),

                                       cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,

                                                                         cdr_ATTACH_REQUEST_WithGuti(v_AttachType,

                                                                                                     v_AdditionalUpdateType,

                                                                                                     ?,

                                                                                                     f_GutiParameters2MobileIdentity(omit, p_GutiOldParams), // GUTI1

                                                                                                     cr_TAI( v_NAS_Old_PlmnId, tsc_TAC2)),

                                                                         cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, ?))))

      -> value v_NasInd;

    v_PdnConnectivityRequest := v_NasInd.Signalling.Nas[0].Pdu.PiggybackedPduList[0].Msg;

    v_EPS_TI := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier; // Use this value in SS responses

    // Get the PCO to be used later (TS 36.508 v.8.2.2 Table 4.7.3-6)

    if (ispresent( v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions)) {

      v_Pco := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions;

    }

    //+   The SS transmits an AUTHENTICATION REQUEST message to initiate the EPS

    //+   authentication and AKA procedure.

    //+   The UE transmits an AUTHENTICATION RESPONSE message and establishes

    //+   mutual authentication.

    //+   The SS transmits a NAS SECURITY MODE COMMAND message to

    //+   activate NAS security.

    //+   The UE transmits a NAS SECURITY MODE COMPLETE message

    //+   and establishes the initial security configuration.

    //+  - ESM information optionally transferred

    //@siclog "Step 5 - 9" siclog@

    v_EIT_Flag := f_CheckEsmInfoTransferFlag(v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST);

    v_APN := f_EUTRA_Authentication_NAS_AS_Security (eutra_CellA, v_EIT_Flag, v_EPS_TI, true);

    //@siclog "Step 10" siclog@

    //+   SS responds with ATTACH ACCEPT message including a valid TAI list. The

    //+   ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message is piggybacked in

    //+   ATTACH ACCEPT message.

    //+   Note: The IP addresses of the UE are not allocated in this test so PDN

    //+   address is not included in the message.

    f_EUTRA_RRCConnectionReconfigurationWithAttachAcceptAllParams(eutra_CellA,

                                                                  v_EPS_Bearer_Id,

                                                                  v_RRC_TI,

                                                                  v_EPS_TI,

                                                                  v_AttachType,

                                                                  f_EUTRA_GetPDNAddress(f_CheckPCOforIPallocationViaNas (v_Pco)),

                                                                  f_GutiParameters2MobileIdentity(tsc_IEI_Guti, p_GutiParams), // GUTI1

                                                                  cs_TrackingAreaIdListDef_lv,

                                                                  v_LAI,

                                                                  v_MSId,

                                                                  omit,

                                                                  v_APN,

                                                                  f_GetDefPCOforPCSCFAddress (v_Pco), // @sic R5-104703 sic@

                                                                  f_GetAdditionalUpdateResult(v_NasInd.Signalling.Nas[0].Pdu.Msg.aTTACH_REQUEST.addUpdateType,

                                                                                              EPS_Only));  // @sic R5s110176, R5-103681 sic@

    //@siclog "Step 11" siclog@

    //+   Check: does the UE transmit an ATTACH COMPLETE message including an

    //+   ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message?

    f_EUTRA_NAS_AttachComplete (eutra_CellA, v_EPS_Bearer_Id );    
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 11"); // TP2

    //@siclog "Step 12 Void" siclog@

    //@siclog "Step 13" siclog@

    //+   The SS releases the RRC connection.

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);

Change 4

Method name
a_ICMPv6_Handler

Reason for change
As per RFC 2461

4.1.  Router Solicitation Message Format

…

   IP Fields:

      Source Address

                     An IP address assigned to the sending interface, or

                     the unspecified address if no address is assigned

                     to the sending interface.

      Destination Address

                     Typically the all-routers multicast address.

…

4.2.  Router Advertisement Message Format

…

   IP Fields:

      Source Address

                     MUST be the link-local address assigned to the

                     interface from which this message is sent.

      Destination Address

                     Typically the Source Address of an invoking Router

                     Solicitation or the all-nodes multicast address.

Hence while sending the Soucre address TTCN should give the link local address.

Also while configuring the routing information; TTCN should configure the link local address of the server to ensure SS is able to route the packet.

Summary of change
Ensure source IP address is a link local address

Source of change
IP_PTC_Main.ttcn

Before change:

  altstep a_ICMPv6_Handler(IP_Socket_Type p_ICMPv6_ServerSocket) runs on IP_PTC

  {

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var IP_Socket_Type v_RemoteSocket;

    var IP_Socket_Type v_LocalSocket;

    var ICMP_Data_Type v_ReqData;

    var ICMP_Data_Type v_RespData;

    var charstring v_ServerIpAddr := p_ICMPv6_ServerSocket.IpAddr.V6.Addr;

    [] IP_SOCK.receive(cr_ICMPv6_DATA_IND(cr_IP_Socket(?, omit)))   /* allow any IP address to deal with multicasts */

      -> value  v_IP_SOCKET_IND

      {

        v_RemoteSocket := v_IP_SOCKET_IND.DATA.ConnectionId.Remote;

        v_LocalSocket := v_IP_SOCKET_IND.DATA.ConnectionId.Local;

        v_ReqData  := v_IP_SOCKET_IND.DATA.Ind.ICMP.RecvFrom;

        v_RespData := f_ICMPv6_GetResponse(v_ReqData);

        if (lengthof(v_RespData) >  0) {

          if (fl_IPv6AddrIsMulticast(v_LocalSocket.IpAddr.V6.Addr)) {

            // Handle a multicast request

            // SRC address is local link address corresponding to the server (SS), i.e. FE80:xxx

            // DST is the remote socket address (as received in the incoming IP packet)

            // TODO - Make sure that the IP address passed to this alstep is the local link address of the server

            v_LocalSocket.IpAddr.V6.Addr  := v_ServerIpAddr;

          }

          IP_SOCK.send(cs_ICMPv6_DATA_REQ(v_LocalSocket, v_RemoteSocket, v_RespData));

        }

      }

  }

After change:

// Addition

  // Anite change

  function fl_ConvertIPv6AddrLinkLocal(charstring p_IPv6Addr) return charstring

  {

    var charstring v_Result:= p_IPv6Addr;

    var O2_Type v_Prefix;

    v_Prefix := str2oct(substr(p_IPv6Addr,0,4));

    if(v_Prefix != 'FF08'O) {

      v_Result := replace ( p_IPv6Addr , 0 , 20 , "FF08:0000:0000:0000:" );

    }

    return v_Result;

  }
// Modification

altstep a_ICMPv6_Handler(IP_Socket_Type p_ICMPv6_ServerSocket) runs on IP_PTC

  {

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var IP_Socket_Type v_RemoteSocket;

    var IP_Socket_Type v_LocalSocket;

    var ICMP_Data_Type v_ReqData;

    var ICMP_Data_Type v_RespData;

    // Anite change

    var charstring v_ServerIpAddr := fl_ConvertIPv6AddrLinkLocal(p_ICMPv6_ServerSocket.IpAddr.V6.Addr);
    [] IP_SOCK.receive(cr_ICMPv6_DATA_IND(cr_IP_Socket(?, omit)))   /* allow any IP address to deal with multicasts */

      -> value  v_IP_SOCKET_IND

      {

        v_RemoteSocket := v_IP_SOCKET_IND.DATA.ConnectionId.Remote;

        v_LocalSocket := v_IP_SOCKET_IND.DATA.ConnectionId.Local;

        v_ReqData  := v_IP_SOCKET_IND.DATA.Ind.ICMP.RecvFrom;

        v_RespData := f_ICMPv6_GetResponse(v_ReqData);

        if (lengthof(v_RespData) >  0) {

          if (fl_IPv6AddrIsMulticast(v_LocalSocket.IpAddr.V6.Addr)) {

            // Handle a multicast request

            // SRC address is local link address corresponding to the server (SS), i.e. FE80:xxx

            // DST is the remote socket address (as received in the incoming IP packet)

            // TODO - Make sure that the IP address passed to this alstep is the local link address of the server

            v_LocalSocket.IpAddr.V6.Addr  := v_ServerIpAddr;

          }

          IP_SOCK.send(cs_ICMPv6_DATA_REQ(v_LocalSocket, v_RemoteSocket, v_RespData));

        }

      }

  }

Change 5

Method name
fl_Icmpv6Server_Start

Reason for change
IPv6 address in the routing table should be the link local address as per change 4.

Summary of change
Ensure IP address added to the mapping is link local

Source of change
IP_PTC_Main.ttcn

Before change:

  function fl_Icmpv6Server_Start(IP_AddrInfo_Type p_LocalIpAddr,

                                 IP_DrbInfo_Type p_DrbInfo,

                                 boolean p_AllowBroadcast := false) runs on IP_PTC return IP_Connection_Type

  {

    var IP_Connection_Type v_IcmpServerInfo;

    // Bind ICMPv6 socket

    v_IcmpServerInfo := fl_ICMPv6_Bind(p_LocalIpAddr, p_AllowBroadcast);

    // Modify routing table

    vc_IP_PTC_DRB_RoutingTable := f_RoutingTable_AddEntry(vc_IP_PTC_DRB_RoutingTable, v_IcmpServerInfo, p_DrbInfo);

    f_IP_PTC_SendRoutingTable(vc_IP_PTC_DRB_RoutingTable);

    return  v_IcmpServerInfo;

  }

After change:

  function fl_Icmpv6Server_Start(IP_AddrInfo_Type p_LocalIpAddr,

                                 IP_DrbInfo_Type p_DrbInfo,

                                 boolean p_AllowBroadcast := false) runs on IP_PTC return IP_Connection_Type

  {

    var IP_Connection_Type v_IcmpServerInfo;

    // Bind ICMPv6 socket

    v_IcmpServerInfo := fl_ICMPv6_Bind(p_LocalIpAddr, p_AllowBroadcast);

    // Anite change

    // Make sure IP in mapping table is link local

    v_IcmpServerInfo.Local.IpAddr.V6.Addr := fl_ConvertIPv6AddrLinkLocal(p_LocalIpAddr.V6.Addr);
    // Modify routing table

    vc_IP_PTC_DRB_RoutingTable := f_RoutingTable_AddEntry(vc_IP_PTC_DRB_RoutingTable, v_IcmpServerInfo, p_DrbInfo);

    f_IP_PTC_SendRoutingTable(vc_IP_PTC_DRB_RoutingTable);

    return  v_IcmpServerInfo;

  }

Change 6

Test case name
fl_TC_10_4_1_Body 

Reason for change
DRB routing table mapping not created for DRB 2 to handle IP user plan data.

Also delay should be added after 2nd default EPS bearer activation to ensure TTCn allows ICMP signalling.

Note: Similar changes required in following test cases:

10.5.1, 10.5.3, 10.6.1, 10.7.5

Summary of change
Created mapping for DRB 2 and removed mapping for DRB1.

Source of change
ESM_Testcases.ttcn

Before change:

  function fl_TC_10_4_1_Body() runs on EUTRA_PTC

  {

    …

    // The SS transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message with IE EPS Bearer Identity set to new EPS bearer context.

    // Note: the SS allocates a PDN address of a PDN type which is compliant with the PDN type requested by the UE.

    //@siclog "Step 3" siclog@

    v_PDN_Address2 := f_EUTRA_GetPDNAddress (f_CheckPCOforIPallocationViaNas (v_Pco), false, true); // @sic R5s110011 sic@

    f_EUTRA_TxActivateDefaultEpsBearerCtxtReq_DrbReconfig (eutra_CellA,

                                                           tsc_RRC_TI_Def,

                                                           v_EpsBearerId2, // Default bearer ID to additional PDN

                                                           v_PTId_UE,  // use value as defined by the UE

                                                           v_AccessPointName,

                                                           v_PDN_Address2,

                                                           f_GetDefPCOforPCSCFAddress(v_Pco));

    // The UE transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 4" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActivateDefEpsBearerContextAccept(v_EpsBearerId2))));

    …

    // The UE transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 24" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActivateDefEpsBearerContextAccept(v_EpsBearerId2))));

    …

    // The UE transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 37" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActivateDefEpsBearerContextAccept(v_EpsBearerId2))));

    …

  } // function fl_TC_10_4_1_Body  

After change:

  function fl_TC_10_4_1_Body() runs on EUTRA_PTC

  {

    …

    // The SS transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message with IE EPS Bearer Identity set to new EPS bearer context.

    // Note: the SS allocates a PDN address of a PDN type which is compliant with the PDN type requested by the UE.

    //@siclog "Step 3" siclog@

    v_PDN_Address2 := f_EUTRA_GetPDNAddress (f_CheckPCOforIPallocationViaNas (v_Pco), false, true); // @sic R5s110011 sic@

    // Anite change

    // Stop handling on DRB1

    f_DHCPv4_Stop(IP);         /* @sic R5s100135: stop DHCP and routing of DRB1 to the IP PTC sic@ */

    f_ICMPv6_Stop(IP);         /* @sic R5-106675 IP and IMS support sic@ */

    // Start handling on DRB2

    f_DHCPv4_Start(IP, px_IPv4_RemoteAddress, cs_DrbInfo_EUTRA(eutra_CellA, tsc_DRB2));

    f_ICMPv6_Start(IP, px_IPv6_RemoteAddress, cs_DrbInfo_EUTRA(eutra_CellA, tsc_DRB2));
    f_EUTRA_TxActivateDefaultEpsBearerCtxtReq_DrbReconfig (eutra_CellA,

                                                           tsc_RRC_TI_Def,

                                                           v_EpsBearerId2, // Default bearer ID to additional PDN

                                                           v_PTId_UE,  // use value as defined by the UE

                                                           v_AccessPointName,

                                                           v_PDN_Address2,

                                                           f_GetDefPCOforPCSCFAddress(v_Pco));

    // The UE transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 4" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActivateDefEpsBearerContextAccept(v_EpsBearerId2))));

    // Anite change

    f_Delay(f_EUTRA_SetTimerToleranceMax(eutra_CellA, nonProtocolTimer, 1.2));     /* 1.2s delay acc. to 36.508 cl. 4.5A.1 to allow user-plane signalling */
    …

    // The UE transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 24" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActivateDefEpsBearerContextAccept(v_EpsBearerId2))));

    // Anite change

    f_Delay(f_EUTRA_SetTimerToleranceMax(eutra_CellA, nonProtocolTimer, 1.2));     /* 1.2s delay acc. to 36.508 cl. 4.5A.1 to allow user-plane signalling */
    …

    // The UE transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 37" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActivateDefEpsBearerContextAccept(v_EpsBearerId2))));

    // Anite change

    f_Delay(f_EUTRA_SetTimerToleranceMax(eutra_CellA, nonProtocolTimer, 1.2));     /* 1.2s delay acc. to 36.508 cl. 4.5A.1 to allow user-plane signalling */
    …

  } // function fl_TC_10_4_1_Body  

